

Data Validation & Combo Box (VBA)

Data Validation List မှာ ရှေ့ကစာလုံးရိုက်ပြီး ရှာလိုရအောင် လုပ်နည်းလေးပါ။

အမေးများပေမယ့် ကျွန်တော် အမြဲ ရှောင်နေခဲ့တာလေးပါ။ VBA ကို သိပ်မကြွမ်းကျင်ခြင်း ရှင်းပြရခက်ခြင်းကြောင့် ရှင်းမပြဖြစ်ခဲ့တာပါ။ ကျွန်တော့် အလုပ်မှာ Format တွေအကုန်ပြန်ဆွဲတော့ Account Name တွေ Customer list တွေကို Data Validation list တွေနဲ့သာ ပြုလုပ်ပြီး ရိုက်မထည့်စေပဲ ရွေးခိုင်းတယ်။ အဲဒါ တပည့်တွေက ညီးကြပါပြီး။ ဟုတ်တော့လဲ ဟုတ်ပါတယ် Account Name တွေကလဲ 300 ကျော် Customer list ဆိုတာလဲ သိတဲ့အတိုင်းပဲ အများကြီး A to Z စီထားပေမယ့်လဲ တစ်ခုခြင်း လိုက်ရွေးဖို့ဆိုတာ အရမ်း အာရုံစိုက်ရတယ်။ တပည့်တွေအဆင်မပြေဘူးဆိုတော့ ကိုယ့်တပည့်တွေကို အမြဲအာရုံစိုက်တတ်တဲ့ သူဖြစ်တဲ့အတွက် မနေနိုင်ဘူး ဖြေရှင်းနိုင်ဖို့ ကြိုးစားရပြီပေါ့။ ဒီလိုနဲ့ လိုက်ရှာရင်း ဒီအသုံးပြုပုံကို တွေ့ရှိတဲ့အတွက် စမ်းသုံးကြည့်ရာမှာ အဆင်ပြေခဲ့တယ်။



နောက်ပိုင်းမှာ မေးလာတာတွေ အရမ်းများလာတဲ့အတွက် အားလုံး အဆင်ပြေပါစေဆိုပြီး ရေးပေးလိုက်ပါတယ်။

ကဲစလိုက်ရအောင် အောက်က ပုံမှာ တွေ့ရတဲ့အတိုင်း ကျွန်တော်က Cash Book ကို Account Name ရွေးလိုက်တာနဲ့ Account Code, Dept Code, Account Group နဲ့ Account Type ပေါ်လာအောင် Index, Match နဲ့ ချိတ်ထားတယ်။ ဟီး ထည့်ကြားလိုက်ဦးမယ် နောက်မှာ Cash Flow နဲ့ Cash Trail ထပ်ချိတ်ထားပြီး နောက် Ledger နဲ့ Financial Report Folder ထဲ ကူးထွဲလိုက်တာနဲ့ ဂျာနယ်အနည်းငယ်ရိုက်ပြီး Final Report ထွက်လာအောင် လုပ်ထားတာ။


					2,997,100	10,870,000	(7,872,900)
Account Code	Dept	Account Name	Account Group	Account Type	Dr	Cr	Balance
		Opening Balance			156,000,000		156,000,000
90020	1000	Data Processing Exp	A&G	Expenses		250,000	155,750,000
10008	100	Cash at Bank	Current Assets	Assets		10,000,000	145,750,000
10004	100	Account Receivable Other	Current Assets	Assets	255,000		146,005,000
90010	1000	Main-Vehicle (A&G)	A&G	Expenses		10,000	145,995,000
20001	200	Account Payable	Current Liabilities	Liabilities		500,000	145,495,000
90019	1000	Bonus & Allowance	A&G	Expenses	2,200,000		147,695,000
90010	6000	Bonus & Allowance	tail Car	Expenses		100,000	147,595,000
20011	200	Data Processing Exp	Current Liabilities	Liabilities	10,000		147,605,000
20011	200	Donation	Current Liabilities	Liabilities		10,000	147,595,000
90015	8000	Education Allowance	le & Marketing	Expenses	20,000		147,615,000
50002	500	Gift away/ Present	Sale Account	Income	100,000		-
50007	500	Insurance	Sale Account	Income	50,000		-
		Licence & Tax					
		Man Power Development					
		Sale-branch					
		Sale-Retail Car					


Data Validation list ဟာ များတဲ့အတွက် လိုက်ရှာရတာ အခက်အခဲဖြစ်စေတယ် ဒီအခါ Data Validation list Drop Down Box မှာ အရှေ့ကစာလုံး ရိုက်ရှာလိုရတဲ့ Combo Box ကို အစားသွင်းအသုံးပြုမှာဖြစ်တယ်။ Data Validation list ပြုလုပ်ပုံကိုတော့ အားလုံးသိပြီးသား ဖြစ်မယ်လို့ ယူဆတဲ့အတွက် ထပ်မရှင်းပြတော့ပါဘူး။ မသိဘူးဆိုလဲ ကျွန်တော်ရေးဘူးတဲ့ စာတွေရှာဖတ်ကြပေးပါ။

Developer Tab ကို နှိပ်ပါ။ Developer Tab မရှိရင် အောက်ကပုံပါအတိုင်း FILE -> Excel Option -> Customize Ribbon မှာ Developer ကို အမှန်ဖြစ်လေး ထည့်ပေးပါ။ ပြီးလျှင် OK နှိပ်ပါ။ ဒါဆို Developer Tab ကိုတွေ့ ရပါလိမ့်မယ်။ Developer Tab -> Controls Group က Insert မှာ Combo Box (ActiveX Control) ကိုနှိပ်ပါ။



ဒါဆိုရင် ကျွန်တော်တို့ Combo Box တစ်ခု ဆွဲလို့ရပါပြီ မိမိ အဆင်ပြေတဲ့ နေရာတစ်နေရာမှာ Combo Box တစ်ခုကို ရေးဆွဲလိုက်ပါ။ ပြီးနောက် အဆိုပါ Combo Box ကို Right Click နှိပ်ပြီး Properties ကိုရွေးပါ Combo Box Properties Window ကျလာပါလိမ့်မယ်။





ဒီနေရာမှာ Combo Box Name ကို TempCombo လို့ အမည်ပေးပါ။ ဘာကြောင့်လဲဆိုတော့ နောက်ပိုင်း VBA code တွေမှာ Combo Box Name နေရာမှာ TempCombo လို့ သုံး ထားလို့ပါ အခြားအမည် ပေးမယ်ဆိုရင် VBA Code တွေမှာ ပြန်ပြင်ရပါမယ်။

Font နေရာမှာတော့ မိမိ အသုံးပြုမယ့် Font အမျိုးအစား အရွယ်အစားကို Font Dialog Box မှာ ဝင်ပြင်ပါ။



ListRows မှာ Drop Down list မှာ မိမိမြင်လိုတဲ့ အရေအတွက်ကို ပြောင်းပါ ကျွန်တော်ကတော့ Default ပါတဲ့အတိုင်း 8 ခုကိုရွေးထားပါတယ်။

MatchEntry မှာ Auto Complete ဖြစ်အောင် 1-fmMatchEntryComplete ကိုရွေးပေးပါ။

ကျွန်တာတွေကို ဒီအတိုင်းထားပြီး ပြင်သင့်တာတွေ ပြင်ပြီးသွားပြီဆို Properties Window ကိုပိတ်ပါ။ ပြီးလျှင် Developer Tab မှာ Design Mode On နေတာကို တစ်ချက်နှိပ်ပြီး Off ပေးပါ။ Design Mode On ထားခြင်းဖြင့် Combo Box ကို လိုအပ်သလို နေရာရွှေ့ခြင်း အရွယ်အစားပြောင်းလဲ ခြင်းများကို ပြုလုပ်နိုင်ပါတယ်။ Design Mode Off ထားမှသာ Combo ကို အသုံးပြုလို့ရပါတယ်။ ဒီမှာတော့ Design Mode Off တာ ဟာ VBA Code တွေ ရေးပြီးသွားတဲ့အခါ Excel Worksheet မှာ ပေါ်နေတဲ့ Combo Box ပျောက်သွားအောင်ဖြစ်ပြီး Code ရေးပြီးမှ Off လဲ ရပါတယ်။

ကဲဒါဆို ကျွန်တော်တို့ Data Validation list ရှိတဲ့ Cell တွေမှာ Double Click နှိပ်လိုက်တာနဲ့ Combo Box ပေါ်လာပြီး ရှေ့စာလုံးရိုက်ပြီး ရှာလို့ရအောင် VBA Code ရေးပါမယ်။ Developer Tab က Controls Group မှာ View Code ကို နှိပ်ပါ။ Microsoft Visual Basic for Applications Window ပွင့်လာရင် ကိုယ် VBA Code ရေးမယ့် Worksheet ကို Double Click နှိပ်ပြီး ရွေးပါ။ နဂိုကတည်းက အဲ့ Worksheet မှာ ရှိနေတယ်ဆိုရင်တော့ တစ်ခါထဲ ဖွင့်ပြီးသား ဖြစ်နေမှာပါ။ အဆိုပါ Sheet Module မှာ အောက်က VBA code ကိုရေးပါတယ်။ ပြီးလျှင် အောက်ပုံပါအတိုင်း File -> Close and Return to Microsoft Excel ကို ရွေးလိုက်ပါ။ Short cut က ဘေးမှာ မြင်နေရတဲ့အတိုင်း Alt+Q ပါပဲ။ ဒါဆို ကျွန်တော်တို့ Data Validation Drop Down list ရှိတဲ့နေရာမှာ Double Click နှိပ်လိုက်တာနဲ့ Combo Box ပေါ်လာပြီး သက်ဆိုင်ရာ စာရိုက်ပြီး ရှာလို့ ရပြီပေါ့။

```
Private Sub Worksheet_BeforeDoubleClick _  
    (ByVal Target As Range, _  
        Cancel As Boolean)  
    Dim str As String  
    Dim cboTemp As OLEObject  
    Dim ws As Worksheet  
    Set ws = ActiveSheet  
    Set cboTemp = ws.OLEObjects("TempCombo")  
    On Error Resume Next  
    With cboTemp  
        .ListFillRange = ""  
        .LinkedCell = ""  
        .Visible = False  
    End With  
    On Error GoTo errHandler  
    If Target.Validation.Type = 3 Then  
        Cancel = True  
        Application.EnableEvents = False  
        str = Target.Validation.Formula1  
        str = Right(str, Len(str) - 1)  
        With cboTemp  
            .Visible = True  
            .Left = Target.Left  
            .Top = Target.Top  
            .Width = Target.Width + 1  
            .Height = Target.Height + 1  
            .ListFillRange = str  
            .LinkedCell = Target.Address  
        End With  
        cboTemp.Activate  
        Me.TempCombo.DropDown  
    End If  
  
errHandler:  
    Application.EnableEvents = True  
    Exit Sub  
  
End Sub  
Private Sub TempCombo_LostFocus()  
    With Me.TempCombo  
        .Top = 10  
        .Left = 10  
        .Width = 0  
        .ListFillRange = ""  
        .LinkedCell = ""  
        .Visible = False  
        .Value = ""  
    End With  
End Sub
```



Code တွေရေးထားတာကို တစ်ခုခြင်း ပြန်ကြည့်ကြရအောင်။ ဒီမှာတော့ ဘာကြောင့်ရေးတယ်ဆိုတာနဲ့ ဘယ်လိုအလုပ်လုပ်တယ်ဆိုတာပဲ ပြောပြထားခြင်းသာ ဖြစ်ပါတယ်။

Double Clicked မနိပ်ခင်မှာ Combo Box နဲ့ သူ့မှာပါတဲ့ List တွေပေါ်မနေအောင် Code တွေရေးပါမယ်။ ပထမ ရေးမယ့် Code ထဲမှာ ပါမယ့် အရာ တွေကို အရင်သတ်မှတ်ပါတယ်။ ဒီမှာ သတိထားရမှာက မိမိ Combo Box ကိုပေးခဲ့တဲ့နာမည်ဟာ ခုရေးမှာနဲ့ တူနေရပါမယ်။

```
Set ws = ActiveSheet
```

```
Set cboTemp = ws . OLEObjects("TempCombo")
```

နောက်တော့ Double Clicked မနိပ်ခင်မှာ Combo Box ပျောက်နေအောင်နဲ့ linked cell နဲ့ listFillRange တွေမှာ Default အတိုင်း ဘာမှ ဝင်မနေအောင် အောက်ပါအတိုင်း Code ထပ်ရေးပါတယ်။ (linked cell နဲ့ listFillRange တွေရဲ့ အသုံးတွေကို Combo Box သုံးဘူးသူတိုင်း သိပါတယ်။ နောက်မှ Combo Box အကြောင်းအသေးစိတ် ထပ်ရေးပေးပါမယ်)

```
With cboTemp
```

```
.ListFillRange = ""
```

```
.LinkedCell = ""
```

```
.Visible = False
```

```
End With
```

နောက်ထပ် Target Cell (Active Cell) ရဲ့ Data Validation Type ဟာ 3 ဖြစ်မဖြစ် ကြည့်ခိုင်းဖို့ Code ထပ်ရေးပါတယ်။ ဒီမှာ Data Validation Type တွေကို ရှင်းပြရရင် Allow မှာ Down Arrow ကိုနှိပ်တဲ့အခါ မြင်ရတဲ့အတိုင်း Any Value ကိုဖယ်ပြီး Whole number ကနေ တစ်နှစ်သုံးရေတွက်တာဖြစ်ပါတယ် အဲ့တော့ Data Validation List ဟာ Type 3 ဖြစ်ပါတယ်။ တစ်ကယ်လို့ Active Cell ဟာ Drop Down list မဟုတ်ရင် Code ကို မ run စေပဲ Drop Down list ဖြစ်မှသာ Code run ရန်အတွက်

```
On Error GoTo errHandler
```

```
If Target.Validation.Type = 3 Then
```

```
Cancel = True
```

```
Application.EnableEvents = False
```

နောက်တစ်ခါ str function အတွက် formula ရေးပါတယ်။ (Str Function ဟာ VBA မှာ အသုံးပြုတဲ့ Excel Function တစ်ခုဖြစ်ပါတယ်။) ကျွန်တော်တို့ Data Validation မှာ Name range တစ်ခုနဲ့ ပေးထားတယ်ဆို ရှေ့က = ထည့်ပြီး ရေးရပါတယ်။ ဥပမာ ခုကျွန်တော်က COA မှာ Account Name တွေကို ACCN လို့ ပေးထားတဲ့ အတွက် Source မှာ =ACCN လို့ရေးထားပါတယ်။ အဲ့မှာ = ကိုဖြုတ်ပြီး ACCN ဆိုတဲ့ Range Name ကိုသာ လိုချင်တဲ့အတွက် Right Function အသုံးပြုထားပါတယ်။

```
str = Target.Validation.Formula1
```

```
str = Right(str, Len(str) - 1)
```

Target Cell (Active Cell) ဟာ Data Validation list ဖြစ်နေပြီးဆိုရင် Combo Box ကို Active Cell ရဲ့ Top left မှာ ပေါ်လာအောင် နောက် Combo Box ဟာ Active Cell နဲ့ Size တူညီအောင် အောက်ပါအတိုင်း Code ရေးပါတယ်။ ဒီမှာ နောက်က ပေါင်းထားတဲ့ + 1 ဟာ Combo Box ကို Active Cell ထပ် အနည်းငယ် ကြီးနေအောင် ပေါင်းထားတာ ဖြစ်ပါတယ်။ မိမိစိတ်ကြိုက် အရွယ်အစားရအောင် ပြောင်းလဲ လို့ ရပါတယ်။

```
With cboTemp
```

```
.Visible = True
```

```
.Left = Target.Left
```

```
.Top = Target.Top
```



```
.Width = Target.Width + 1
```

```
.Height = Target.Height + 1
```

ထပ်ပြီး Combo Box properties ရှိ ListFillRange မှာ str Variable ဖြစ်တဲ့ ACCN ဆိုတဲ့ Range သွားဝင်အောင်

```
.ListFillRange = str
```

နောက် LinkedCell ဟာလဲ ကိုယ်လက်ရှိ Select လုပ်ထားတဲ့ Cell ဖြစ်သွားအောင်

```
.LinkedCell = Target.Address
```

လို့ ရေးပါတယ်။ Combo Box Active ဖြစ်လာအောင်နဲ့ Drop down list ပွင့်လာအောင် နောက်ဆုံးမှာ အောက်ပါ Code ကို ရိုက်ပါတယ်။

```
cboTemp.Activate
```

```
Me.TempCombo.DropDown
```

```
End If
```

အပေါ်က Code တွေဟာ Combo Box Drop Down list ပေါ်လာအောင် ရေးထားတာဖြစ်ပါတယ်။

နောက်ထပ်ပြီးတော့ ရွေးပြီးသွားတဲ့အခါမှာ Combo Box ပျောက်သွားအောင်နဲ့ linked cell နဲ့ ListFillRange မှာ clear ဖြစ်သွားအောင် အောက်ပါအတိုင်း Code ထပ်ရေးပါတယ်။

```
Private Sub TempCombo_LostFocus()
```

```
With Me.TempCombo
```

```
.Top = 10
```

```
.Left = 10
```

```
.Width = 0
```

```
.ListFillRange = ""
```

```
.LinkedCell = ""
```

```
.Visible = False
```

```
.Value = ""
```

```
End With
```

```
End Sub
```

ကဲ ဒီလိုဆိုရင် အခု ရေးထားတဲ့ Code တွေ ဘာကြောင့်ရေးထားတယ်ဆိုတာ သိကြပြီ ထင်ပါတယ်။ ဒီမှာ Drop Down list ကို အစ စကားလုံးရိုက်ပြီး ရှာလိုရအောင် လုပ်ပြီးသွားပြီပေါ့။ နောက် ပြဿနာတစ်ခု ကျန်နေသေးတယ် အဲ့ဒါကတော့ စာလုံးရွေးပြီး ရိုက်ပြီးသွားရင် Enter ခေါက်ပြီး Tab ရိုက်ပြီး Active Cell ကို ရွေ့လို့မရပဲ Esc နှိပ်ပြီးမှသာ ရွေ့လို့ ရနေတာလေး ပြင်ရမှာပေါ့။

နောက်ထပ် Code တစ်ခု ထပ်ရေးပါတယ်။

```
Private Sub TempCombo_KeyDown(ByVal _
```

```
KeyCode As MSForms.ReturnInteger, _
```

```
ByVal Shift As Integer)
```

```
Select Case KeyCode
```

```
Case 9
```

```
ActiveCell.Offset(0, 1).Activate
```

Case 13

ActiveCell.Offset(1, 0).Activate

Case Else

End Select

End Sub

ဒီ Code တွေကတော့ ဖတ်လိုက်တာနဲ့ နားလည်နိုင်တာ ဖြစ်တဲ့အတွက် မရှင်းတော့ ပါဘူး။ ဗဟုသုတအနေနဲ့ Case KeyCode တွေမှာ 9 ဆိုတာ Tab ဖြစ်ပြီး 13 ဆိုတာ Enter ဖြစ်ပါတယ်။ ဒီလောက်ဆိုရင်တော့ Data Valadation list နဲ့ Combo Box တွဲ ပြီး များပြားလှတဲ့ List တွေကို ရှေ့ကစကားလုံး ရိုက်ရုံနဲ့ ရှာဖွေရေးသွင်းနိုင်ပါပြီ။ နောက်ထပ်ပြီး အကယ်၍ list ကို Indirect Function သုံးထားတယ် ဆိုရင်တော့ အပေါ်ကရေးထားတဲ့ အောက်က Code အောက်မှာ အောက်ပါအတိုင်း အစားသွင်းရမှာ ဖြစ်ပါတယ်။

str = Target.Validation.Formula1

str = Right(str, Len(str) - 1)

If Left(str, 4) = "INDI" Then

ISplit = InStr(1, str, "(")

str = Right(str, Len(str) - ISplit)

str = Left(str, Len(str) - 1)

str = Range(str).Value

End If

ဒီလောက်ဆိုရင်တော့ Drop Down list မှာ ရှေ့စကားလုံးရိုက်ပြီး ရှာဖွေရေးသွင်းတဲ့ နည်းလမ်းနဲ့ ပြုလုပ်ပုံ အဆင့်ကို သိရှိနားလည်ပြီး ကိုယ်တိုင် ပြုလုပ်နိုင်လောက်ပြီလို့ ယုံကြည်ပါတယ်။ Code တွေကိုလဲ အလွယ်တစ်ကူ ကူးယူလို့ရအောင် ပုံမှာတွဲပြီး ရေးပေးထားပါတယ်။ အားလုံးအဆင်ပြေကြပါစေ။ နောက်တစ်ခုပြောလိုတာက Excel File ကို Save လုပ်တဲ့အခါ File -> Save as ကနေ Excel Macro – Enabled Workbook အနေနဲ့ Save ဖို့တော့ လိုပါလိမ့်မယ်။

ဆရာသမားများအား ဦးညွှတ်လျက် ညွှန်လွှင့်ထွန်း

24/12/16 (01:15 AM)